
클라우드 리소스 프로비저닝 경험이의 영역 내에만 있는 경우 AWS Cloud Development Kit 
(AWS CDK) 및 이외의 코드형 인프라(IaC) 도구에 대한 경험이 제한적일 AWS수 있습니다AWS 
CloudFormation. 사실 Hashicorp Terraform과 같은 유사한 도구는 완전히 익숙하지 않을 수 있습니다. 
그러나 클라우드 여정에 더 깊이 들어갈수록 Terraform이 더 피할 수 없게 됩니다. 핵심 개념을 숙지하
는 것이 유리할 것입니다.

Terraform, AWS CDK및 CloudFormation은 유사한 목표를 달성하고 많은 핵심 개념을 공유하지만 몇 
가지 차이점이 있습니다. Terraform에 처음 접근하는 경우 이러한 차이에 대비하지 못할 수 있습니다. 
결국 AWS CDK 및 CloudFormation 스택은 모두 내부에 기반 AWS 계정하므로 유지 관리하는 대부분
의 리소스와 직접적인 관계가 있습니다. Terraform은 단일 클라우드 공급자의 환경 내에 기반하지 않
습니다. 이렇게 하면 다양한 공급자를 유연하게 지원할 수 있지만 원격 위치에 대한 리소스를 유지해야 
합니다.

이 가이드는 Terraform의 핵심 개념을 설명하여 원하는 IaC 문제를 처리하는 데 도움이 됩니다. 
Terraform이 공급자, 모듈 및 상태 파일과 같은 개념을 사용하여 리소스를 프로비저닝하는 방법에 중점
을 둡니다. 또한 Terraform 개념과 AWS CDK 및 CloudFormation이 유사한 작업을 수행하는 방법을 대
조합니다.

Note

는 개발자가 프로그래밍 방식의 코딩 언어를 사용하여 CloudFormation 스택을 배포하는 데 
AWS CDK 도움이 됩니다. 를 실행하면 cdk synth코드가 CloudFormation 템플릿으로 변환
됩니다. 이 시점부터 프로세스는 AWS CDK 와 CloudFormation 간에 동일합니다. 간결성을 위
해이 가이드는 일반적으로 CloudFormation 용어로 AWS IaC 프로세스를 참조하지만 비교는에 
대해서도 마찬가지입니다 AWS CDK.

CloudFormation 및 Terraform 용어

Terraform을 AWS CDK 및 CloudFormation과 비교할 때 IaC 핵심 개념을 설명하는 데 사용되는 용어
가 일관되지 않기 때문에 IaC 핵심 개념을 조정하는 것이 어려울 수 있습니다. 다음은 이러한 용어와이 
가이드에서 이러한 용어를 참조하는 방법입니다.

• 스택 -  스택은 CI/CD 파이프라인에 배포되고 단일 단위로 추적 가능한 IaC입니다.  이 용어는 
CloudFormation에서 일반적이지만 Terraform은 실제로이 용어를 사용하지 않습니다. Terraform 스
택은 모든 하위 모듈이 있는 배포된 루트 모듈입니다. 그러나 모듈이라는 용어와 혼동을 피하기 위해
이 가이드에서는 스택이라는 용어를 사용하여 두 도구 모두에 대한 단일 배포를 설명합니다.

• 상태 - 상태는 모두 현재 추적되는 리소스이며 IaC 배포 스택 내에서 현재 구성입니다. Terraform 상
태 및 백엔드 이해 섹션에 설명된 대로 Terraform은 CloudFormation보다 상태라는 용어를 더 많이 
사용합니다. 이는 상태를 유지하는 것이 Terraform에서 더 잘 보이지만 상태를 추적하고 업데이트하
는 것도 CloudFormation에서 똑같이 중요하기 때문입니다.

• IaC 파일 -  IaC 파일은 코드형 인프라(IaC) 언어를 포함하는 단일 파일입니다. CloudFormation은 단
일 CloudFormation 파일을 템플릿이라고 합니다. 그러나 Terraform의 템플릿과 템플릿 파일은 완전
히 다릅니다. Terraform의 CloudFormation 템플릿과 동일한를 구성 파일이라고 합니다. 이 설명서에
서 혼동을 최소화하기 위해 파일 또는 IaC 파일이라는 용어는 CloudFormation 템플릿과 Terraform 
구성 파일을 모두 참조하는 데 사용됩니다.

다음 표에서는 CloudFormation과 Terraform에 사용되는 용어를 비교합니다. 이 표의 의도는 유사성을 
보여주는 것입니다. 이는 one-to-one 비교가 아닙니다. 각 개념은 CloudFormation과 Terraform 간에 최
소한 약간 다릅니다. 개념은이 가이드의 관련 섹션에 자세히 설명되어 있습니다.

CloudFormation 용어 Terraform 용어 이 가이드의 섹션

CDK 인터페이스(예:IBucket) 데이터 소스 테라폼 데이터 소스 이해

변경 세트 계획 테라폼 모듈에 대한 이해

조건 함수 조건식 Terraform 함수, 표현식 및 메
타 인수 이해

DependsOn  속성 depends_on  메타 인수 Terraform 함수, 표현식 및 메
타 인수 이해

CloudFormation 및 Terraform 용어 2

CloudFormation 용어 Terraform 용어 이 가이드의 섹션

내장 함수 함수 Terraform 함수, 표현식 및 메
타 인수 이해

모듈 모듈 테라폼 모듈에 대한 이해

출력 출력 값 Terraform 변수, 로컬 값 및 출
력 이해

파라미터 변수 Terraform 변수, 로컬 값 및 출
력 이해

레지스트리 공급자 테라폼 제공자에 대한 이해

템플릿 구성 파일 모두

CloudFormation 및 Terraform 용어 3



AWS 권장 가이드 Terraform 시작하기: AWS CDK 및 AWS CloudFormation 전문가를 위
한 지침

Terraform 리소스 이해
AWS CloudFormation 및 Terraform이 모두 존재하는 주된 이유는 클라우드 리소스의 생성 및 유지 관
리입니다. 하지만 클라우드 리소스란 정확히 무엇입니까? CloudFormation 리소스와 Terraform 리소스
는 동일한가요? 답은... 예와 아니요입니다. 이를 설명하기 위해이 가이드에서는 CloudFormation을 사
용한 다음 Terraform을 사용하여 Amazon Simple Storage Service(Amazon S3) 버킷을 생성하는 예를 
제공합니다.

다음 CloudFormation 코드 예제에서는 샘플 Amazon S3 버킷을 생성합니다.

{ 
    "myS3Bucket": { 
        "Type": "AWS::S3::Bucket", 
        "Properties": { 
            "BucketName": "my-s3-bucket", 
            "BucketEncryption": { 
                "ServerSideEncryptionConfiguration": [ 
                    { 
                        "ServerSideEncryptionByDefault": { 
                            "SSEAlgorithm": "AES256" 
                        } 
                    } 
                ] 
            }, 
            "PublicAccessBlockConfiguration": { 
                "BlockPublicAcls": true, 
                "BlockPublicPolicy": true, 
                "IgnorePublicAcls": true, 
                "RestrictPublicBuckets": true 
            }, 
            "VersioningConfiguration": { 
                "Status": "Enabled" 
            } 
        } 
    }
}

다음 Terraform 코드 예제에서는 동일한 Amazon S3 버킷을 생성합니다.

resource "aws_s3_bucket" "myS3Bucket" { 
  bucket = "my-s3-bucket"

4

}

resource "aws_s3_bucket_server_side_encryption_configuration" "bucketencryption" { 
  bucket = aws_s3_bucket.myS3Bucket.id 
  rule { 
    apply_server_side_encryption_by_default { 
      sse_algorithm = "AES256" 
    } 
  }
}

resource "aws_s3_bucket_public_access_block" "publicaccess" { 
  bucket                  = aws_s3_bucket.myS3Bucket.id 
  block_public_acls       = true 
  block_public_policy     = true 
  ignore_public_acls      = true 
  restrict_public_buckets = true
}

resource "aws_s3_bucket_versioning" "versioning" { 
  bucket = aws_s3_bucket.myS3Bucket.id 
  versioning_configuration { 
    status = "Enabled" 
  }
}

Terraform의 경우 공급자는 리소스를 정의한 다음 개발자는 해당 리소스를 선언하고 구성합니다. 공급
자는이 가이드에서 다음 섹션에서 설명하는 개념입니다. Terraform 예제에서는 여러 S3 버킷의 설정
에 대해 완전히 별도의 리소스를 생성합니다. 설정을 위한 별도의 리소스를 생성하는 것이 Terraform 
AWS 공급자가 AWS 리소스를 처리하는 방식에 반드시 일반적인 것은 아닙니다. 그러나이 예제는 중
요한 차이점을 보여줍니다. CloudFormation 리소스는 CloudFormation 리소스 사양에 따라 엄격하게 
정의되지만 Terraform에는 이러한 요구 사항이 없습니다. Terraform에서 리소스의 개념은 약간 더 모
호합니다.

단일 리소스가 무엇인지 정의하는 정확한 가드레일과 관련하여 도구가 다를 수 있지만 일반적으로 클
라우드 리소스는 클라우드에 존재하고 생성, 업데이트 또는 삭제할 수 있는 모든 특정 엔터티입니다. 
따라서 관련된 리소스 수에 관계없이 이전 두 예제 모두 내에서 정확히 동일한 설정을 사용하여 정확히 
동일한 사물을 생성합니다 AWS 계정.

테라폼 제공자에 대한 이해
Terraform에서 공급자는 클라우드 제공업체, 타사 도구 및 기타 API와 상호 작용하는 플러그인입니다. 
Terraform을 함께 AWS사용하려면 리소스와 상호 작용하는 AWS 공급자를 사용해야 합니다. AWS

AWS CloudFormation 레지스트리를 사용하여 타사 확장을 배포 스택에 통합한 적이 없다면 Terraform
공급자가 익숙해지는 데 시간이 걸릴 수 있습니다. CloudFormation 가 기본으로 제공되기 때문에 기본
적으로 AWS 리소스 공급자가 이미 있습니다. AWS반면 Terraform에는 단일 기본 공급자가 없으므로 
주어진 리소스의 출처에 대해 아무 것도 가정할 수 없습니다. 즉, Terraform 구성 파일에서 가장 먼저 
선언해야 할 것은 리소스가 정확히 어디로 가고 어떻게 전달되는지입니다.

이러한 구분은 Terraform에는 존재하지 않는 복잡성을 한층 더 가중시킵니다. CloudFormation 하지만 
이러한 복잡성으로 인해 유연성이 향상됩니다. 단일 Terraform 모듈 내에서 여러 공급자를 선언할 수 
있으며, 그러면 생성된 기본 리소스가 동일한 배포 계층의 일부로 서로 상호 작용할 수 있습니다.

이는 다양한 방식으로 유용할 수 있습니다. 제공업체가 반드시 별도의 클라우드 공급자를 위한 것은 아
닙니다. 공급자는 클라우드 리소스의 모든 소스를 대표할 수 있습니다. 아마존 엘라스틱 쿠버네티스 서
비스 (Amazon EKS) 를 예로 들어 보겠습니다. Amazon EKS 클러스터를 프로비저닝할 때 Helm 차트
를 사용하여 타사 확장을 관리하고 Kubernetes 자체를 사용하여 포드 리소스를 관리하는 것이 좋습니
다. Helm과 Kubernetes에는 모두 자체 Terraform 공급자가 있으므로 AWS이러한 리소스를 모두 동시
에 프로비저닝 및 통합한 다음 이들 간에 값을 전달할 수 있습니다.

Terraform의 다음 코드 예제에서 AWS 제공자는 Amazon EKS 클러스터를 생성한 다음 결과 쿠버네티
스 구성 정보가 헬름 및 쿠버네티스 제공자 모두에게 전달됩니다.

terraform { 
  required_providers { 
    aws = { 
      source  = "hashicorp/aws" 
      version = ">= 4.33.0" 
    } 

    helm = { 
      source  = "hashicorp/helm" 
      version = "2.12.1" 
    } 

    kubernetes = { 
      source  = "hashicorp/kubernetes" 
      version = "2.26.0" 

    } 
  } 
  required_version = ">= 1.2.0"
}

provider "aws" { 
  region = "us-west-2"
}

resource "aws_eks_cluster" "example_0" { 
  name     = "example_0" 
  role_arn = aws_iam_role.cluster_role.arn 
  vpc_config { 
    endpoint_private_access = true 
    endpoint_public_access  = true 
    subnet_ids              = var.subnet_ids 
  }
}

locals { 
  host        = aws_eks_cluster.example_0.endpoint 
  certificate = base64decode(aws_eks_cluster.example_0.certificate_authority.data)
}

provider "helm" { 
  kubernetes { 
    host                   = local.host 
    cluster_ca_certificate = local.certificate 
    # exec allows for an authentication command to be run to obtain user 
    # credentials rather than having them stored directly in the file 
    exec { 
      api_version = "client.authentication.k8s.io/v1beta1" 
      args        = ["eks", "get-token", "--cluster-name", 
 aws_eks_cluster.example_0.name] 
      command     = "aws" 
    } 
  }
}

provider "kubernetes" { 
  host                   = local.host 
  cluster_ca_certificate = local.certificate 
  exec { 
    api_version = "client.authentication.k8s.io/v1beta1" 

7

    args        = ["eks", "get-token", "--cluster-name", 
 aws_eks_cluster.example_0.name] 
    command     = "aws" 
  }
}

두 IaC 도구의 경우 공급자와 관련하여 절충점이 있습니다. Terraform은 배포를 주도하는 엔진인 외부
에 위치한 공급자 패키지에 전적으로 의존합니다. CloudFormation 모든 주요 AWS 프로세스를 내부적
으로 지원합니다. 를 사용하면 타사 확장 프로그램을 통합하려는 경우에만 타사 공급자에 대해 걱정할 
필요가 있습니다. CloudFormation 각 접근 방식에는 장단점이 있습니다. 어느 것이 적합한지는 이 가이
드의 범위를 벗어나지만, 두 도구를 모두 평가할 때는 차이점을 기억하는 것이 중요합니다.

Terraform 별칭 사용
Terraform에서는 사용자 지정 구성을 각 공급자에게 전달할 수 있습니다. 그렇다면 동일한 모듈 내에서 
여러 공급자 구성을 사용하려면 어떻게 해야 할까요? 이 경우 별칭을 사용해야 합니다.   별칭은 리소
스별 또는 모듈별 수준에서 사용할 제공자를 선택하는 데 도움이 됩니다. 동일한 제공자의 인스턴스가 
두 개 이상 있는 경우 별칭을 사용하여 기본이 아닌 인스턴스를 정의합니다. 예를 들어 기본 제공자 인
스턴스는 특정 AWS 리전제공자 인스턴스일 수 있지만 별칭을 사용하여 대체 지역을 정의할 수 있습니
다.

다음 Terraform 예제는 별칭을 사용하여 다양한 버킷을 프로비저닝하는 방법을 보여줍니다. AWS 리전
공급자의 기본 지역은 이지만us-west-2, east 별칭을 사용하여 리소스를 프로비저닝할 수 있습니다.
us-east-2

provider "aws" { 
  region = "us-west-2"
}

provider "aws" { 
  alias  = "east" 
  region = "us-east-2"
}

resource "aws_s3_bucket" "myWestS3Bucket" { 
  bucket = "my-west-s3-bucket"
}

resource "aws_s3_bucket" "myEastS3Bucket" { 
  provider = aws.east 
  bucket   = "my-east-s3-bucket"

Terraform 별칭 사용 8

}

이전 예제에서 볼 수 있듯이 provider 메타 인수와 alias 함께 를 사용하면 특정 리소스에 대해 다른 
제공자 구성을 지정할 수 있습니다. 단일 스택에 여러 리소스를 프로비저닝하는 것은 AWS 리전 시작
에 불과합니다. 앨리어싱 공급자는 여러 면에서 매우 편리합니다.

예를 들어, 한 번에 여러 Kubernetes 클러스터를 프로비저닝하는 것은 매우 일반적입니다. 별칭은 
Helm 및 Kubernetes 공급자를 추가로 구성하는 데 도움이 되므로 Amazon EKS 리소스별로 이러한 타
사 도구를 다르게 사용할 수 있습니다. 다음 Terraform 코드 예제는 별칭을 사용하여 이 작업을 수행하
는 방법을 보여줍니다.

resource "aws_eks_cluster" "example_0" { 
  name     = "example_0" 
  role_arn = aws_iam_role.cluster_role.arn 
  vpc_config { 
    endpoint_private_access = true 
    endpoint_public_access  = true 
    subnet_ids              = var.subnet_ids[0] 
  }
}

resource "aws_eks_cluster" "example_1" { 
  name     = "example_1" 
  role_arn = aws_iam_role.cluster_role.arn 
  vpc_config { 
    endpoint_private_access = true 
    endpoint_public_access  = true 
    subnet_ids              = var.subnet_ids[1] 
  }
}

locals { 
  host         = aws_eks_cluster.example_0.endpoint 
  certificate  = base64decode(aws_eks_cluster.example_0.certificate_authority.data) 
  host1        = aws_eks_cluster.example_1.endpoint 
  certificate1 = base64decode(aws_eks_cluster.example_1.certificate_authority.data)
}

provider "helm" { 
  kubernetes { 
    host                   = local.host 
    cluster_ca_certificate = local.certificate 
    exec { 

      api_version = "client.authentication.k8s.io/v1beta1" 
      args        = ["eks", "get-token", "--cluster-name", 
 aws_eks_cluster.example_0.name] 
      command     = "aws" 
    } 
  }
}

provider "helm" { 
  alias = "helm1" 
  kubernetes { 
    host                   = local.host1 
    cluster_ca_certificate = local.certificate1 
    exec { 
      api_version = "client.authentication.k8s.io/v1beta1" 
      args        = ["eks", "get-token", "--cluster-name", 
 aws_eks_cluster.example_1.name] 
      command     = "aws" 
    } 
  }
}

provider "kubernetes" { 
  host                   = local.host 
  cluster_ca_certificate = local.certificate 
  exec { 
    api_version = "client.authentication.k8s.io/v1beta1" 
    args        = ["eks", "get-token", "--cluster-name", 
 aws_eks_cluster.example_0.name] 
    command     = "aws" 
  }
}

provider "kubernetes" { 
  alias                  = "kubernetes1" 
  host                   = local.host1 
  cluster_ca_certificate = local.certificate1 
  exec { 
    api_version = "client.authentication.k8s.io/v1beta1" 
    args        = ["eks", "get-token", "--cluster-name", 
 aws_eks_cluster.example_1.name] 
    command     = "aws" 
  }

}

테라폼 모듈에 대한 이해
코드형 인프라 (IaC) 영역에서 모듈은 재사용을 위해 격리되고 함께 패키징되는 독립형 코드 블록입니
다. 모듈의 개념은 Terraform 개발에서 피할 수 없는 부분입니다. 자세한 내용은 Terraform 설명서의 모
듈을 참조하십시오. AWS CloudFormation 모듈도 지원합니다. 자세한 내용은 AWS 클라우드 운영 및 
마이그레이션 블로그의 AWS CloudFormation 모듈 소개를 참조하십시오.

Terraform의 모듈 간의 주요 차이점은 특수 리소스 유형 () 을 사용하여 CloudFormation 모듈을 가져온
다는 것입니다. CloudFormation AWS::CloudFormation::ModuleVersion Terraform의 모든 구성
에는 루트 모듈이라는 모듈이 하나 이상 있습니다. 기본.tf 파일에 있는 테라폼 리소스 또는 Terraform 
구성 파일의 파일은 루트 모듈에 있는 것으로 간주됩니다. 그러면 루트 모듈이 다른 모듈을 호출하여 
스택에 포함할 수 있습니다. 다음 예제는 오픈 소스 eks 모듈을 사용하여 Amazon Elastic Kubernetes 
Service (Amazon EKS) 클러스터를 프로비저닝하는 루트 모듈을 보여줍니다.

terraform { 
  required_providers { 
    helm = { 
      source  = "hashicorp/helm" 
      version = "2.12.1" 
    } 
  } 
  required_version = ">= 1.2.0"
}

module "eks" { 
  source  = "terraform-aws-modules/eks/aws" 
  version = "20.2.1" 
  vpc_id  = var.vpc_id
}

provider "helm" { 
  kubernetes { 
    host                   = module.eks.cluster_endpoint 
    cluster_ca_certificate = 
 base64decode(module.eks.cluster_certificate_authority_data) 
  }
}

위의 구성 파일에 공급자가 포함되어 있지 않다는 사실을 눈치채셨을 것입니다. AWS 모듈은 독립적이
며 자체 제공자를 포함할 수 있기 때문입니다. Terraform 공급자는 전 세계적이므로 하위 모듈의 공급

자를 루트 모듈에서 사용할 수 있습니다. 하지만 모든 모듈 값에 적용되는 것은 아닙니다. 모듈 내의 다
른 내부 값은 기본적으로 해당 모듈로만 범위가 지정되며 루트 모듈에서 액세스할 수 있으려면 출력으
로 선언해야 합니다. 오픈소스 모듈을 활용하여 스택 내 리소스 생성을 단순화할 수 있습니다. 예를 들
어, eks 모듈은 EKS 클러스터를 프로비저닝하는 데 그치지 않고 완벽하게 작동하는 Kubernetes 환경
을 프로비저닝합니다. eks 모듈 구성이 요구 사항에 맞으면 이를 사용하면 수십 줄의 코드를 추가로 작
성하지 않아도 됩니다.

모듈 호출

Terraform 배포 중에 실행하는 두 가지 기본 Terraform CLI 명령은 테라폼 초기화와 테라폼 적용입니
다. 이 terraform init 명령이 수행하는 기본 단계 중 하나는 모든 하위 모듈을 찾아 종속 항목으로 
디렉터리로 가져오는 것입니다. .terraform/modules 개발 중에 외부 소스 모듈을 새로 추가할 때
마다 명령을 사용하기 전에 다시 초기화해야 합니다. apply Terraform 모듈에 대한 참조를 들으면 이 
디렉터리의 패키지를 참조하는 것입니다. 엄밀히 말하면 코드에서 선언하는 모듈은 호출 모듈이므로 
실제로는 module 키워드가 실제 모듈을 호출하며, 이 모듈은 종속성으로 저장됩니다.

이런 식으로 호출 모듈은 배포 시 교체될 전체 모듈을 보다 간결하게 나타내는 역할을 합니다. 원하는 
기준을 사용하여 스택 내에 자체 모듈을 만들어 리소스를 논리적으로 분리함으로써 이 아이디어를 활
용할 수 있습니다. 단, 이 작업의 최종 목표는 스택 복잡성을 줄이는 것이어야 한다는 점만 기억하세요. 
모듈 간에 데이터를 공유하려면 모듈 내에서 해당 데이터를 출력해야 하기 때문에 모듈에 너무 많이 의
존하면 상황이 지나치게 복잡해질 수 있습니다.

루트 모듈

모든 Terraform 구성에는 적어도 하나의 모듈이 있으므로 가장 많이 다루게 될 모듈인 루트 모듈의 모
듈 속성을 검사하는 데 도움이 될 수 있습니다. Terraform 프로젝트를 작업할 때마다 루트 모듈은 최상
위 디렉터리의 모든 .tf (또는.tf.json) 파일로 구성됩니다. 최상위 terraform apply 디렉토리
에서 실행하면 Terraform은 해당 디렉토리에서 찾은 모든 파일을 실행하려고 시도합니다. .tf 하위 디
렉터리의 모든 파일은 이러한 최상위 구성 파일 중 하나에서 호출되지 않는 한 무시됩니다.

이렇게 하면 코드를 어느 정도 유연하게 구성할 수 있습니다. 단일 프로세스에 여러 파일이 포함될 
수 있기 때문에 Terraform 배포를 파일보다 모듈이라고 부르는 것이 더 정확한 이유이기도 합니다. 
Terraform이 모범 사례로 권장하는 표준 모듈 구조가 있습니다. 하지만 최상위 디렉터리에 .tf 파일을 
넣으면 나머지 파일과 함께 실행됩니다. 실제로 모듈을 실행하면 모듈의 모든 최상위 .tf 파일이 배포
됩니다. terraform apply 그렇다면 Terraform은 어떤 파일을 가장 먼저 실행할까요? 이 질문에 대
한 답은 매우 중요합니다.

Terraform은 초기화 후와 스택 배포 전에 수행하는 일련의 단계가 있습니다. 먼저 기존 구성을 분석한 
다음 종속성 그래프를 생성합니다. 종속성 그래프는 어떤 리소스가 필요하고 어떤 순서로 처리해야 하
는지를 결정합니다. 예를 들어 다른 리소스에서 참조되는 속성을 포함하는 리소스는 종속 리소스보다 
먼저 처리됩니다. 마찬가지로 depends_on 파라미터를 사용하여 명시적으로 종속성을 선언한 리소스
는 지정한 리소스보다 먼저 처리됩니다. 가능한 경우 Terraform은 병렬 처리를 구현하고 비종속 리소스
를 동시에 처리할 수 있습니다. 배포하기 전에 terraform graph 명령을 사용하여 종속성 그래프를 볼 수 
있습니다.

종속성 그래프가 생성된 후 Terraform은 배포 중에 수행해야 할 작업을 결정합니다. 종속성 그래프를 
최신 상태 파일과 비교합니다. 이 프로세스의 결과를 계획이라고 하며 이는 CloudFormation 변경 세트
와 매우 비슷합니다. terraform plan 명령을 사용하면 현재 계획을 볼 수 있습니다.

가장 좋은 방법은 표준 모듈 구조에 최대한 가깝게 유지하는 것이 좋습니다. 구성 파일이 너무 길어서 
효율적으로 관리할 수 없고 논리적 분리가 관리를 단순화할 수 있는 경우 코드를 여러 파일에 분산할 
수 있습니다. 스택을 최대한 효율적으로 실행하기 위해 종속성 그래프 및 계획 프로세스가 어떻게 작동
하는지 염두에 두세요.

Terraform 상태 및 백엔드 이해
코드형 인프라(IaC)에서 가장 중요한 개념 중 하나는 상태 개념입니다. IaC 서비스는 상태를 유지하므
로 배포할 때마다 리소스를 다시 생성하지 않고도 IaC 파일에 리소스를 선언할 수 있습니다. IaC 파일
은 배포가 끝날 때 모든 리소스의 상태를 문서화하므로 다음 배포에서 선언된 대로 해당 상태를 대상 
상태와 비교할 수 있습니다. 따라서 현재 상태에 라는 Amazon Simple Storage Service(Amazon S3) 버
킷이 포함되어 my-s3-bucket 있고 수신되는 변경 사항에 동일한 버킷도 포함되어 있는 경우, 새 프
로세스는 모든 새 버킷을 생성하려고 하지 않고 기존 버킷에 발견된 모든 변경 사항을 적용합니다.

다음 표에는 일반적인 IaC 상태 프로세스의 예가 나와 있습니다.

현재 상태 대상 상태 작업

이름이 인 S3 버킷 없음 my-
s3-bucket

이름이 인 S3 버킷 my-s3-
bucket

라는 S3 버킷 생성 my-s3-
bucket

my-s3-bucket  버킷 버전 관
리가 구성되지 않은 경우

my-s3-bucket  버킷 버전 관
리가 구성되지 않은 경우

작업 없음

my-s3-bucket  버킷 버전 관
리가 구성되지 않은 경우

my-s3-bucket  버킷 버전 관
리가 구성된

버킷 버전 관리를 my-s3-buc 
ket  사용하도록 구성

my-s3-bucket  버킷 버전 관
리가 구성된

이름이 인 S3 버킷 없음 my-
s3-bucket

삭제 시도 my-s3-bucket

AWS CloudFormation 및 Terraform 트랙 상태의 다양한 방법을 이해하려면 두 도구 간의 첫 번째 기
본 차이점을 기억해야 합니다. CloudFormation은 내부에서 호스팅 AWS 클라우드되고 Terraform은 
기본적으로 원격입니다. 이 사실을 통해 CloudFormation은 내부적으로 상태를 유지할 수 있습니다. 
CloudFormation 콘솔로 이동하여 지정된 스택의 이벤트 기록을 볼 수 있지만 CloudFormation 서비스 
자체가 상태 규칙을 적용합니다.

지정된 리소스에 대해 CloudFormation이 작동하는 세 가지 모드는 Create, Update및 입니
다Delete. 현재 모드는 마지막 배포에서 발생한 일을 기반으로 결정되며, 그렇지 않으면 영향을 받을 
수 없습니다. 결정된 모드에 영향을 미치기 위해 CloudFormation 리소스를 수동으로 업데이트할 수 있
지만 "이 리소스의 경우 Create 모드에서 운영"이라는 명령을 CloudFormation에 전달할 수 없습니다.

Terraform은에서 호스팅되지 않으므로 상태를 유지하는 AWS 클라우드프로세스를 더 구성할 수 있어
야 합니다. 이러한 이유로 Terraform 상태는 자동으로 생성된 상태 파일 내에서 유지됩니다. Terraform 

개발자는 CloudFormation보다 훨씬 더 직접적으로 상태를 처리해야 합니다. 기억해야 할 중요한 점은 
추적 상태가 두 도구 모두에 동일하게 중요하다는 것입니다.

기본적으로 Terraform 상태 파일은 Terraform 스택을 실행하는 기본 디렉터리의 최상위에 로컬로 저
장됩니다. 로컬 개발 환경에서 terraform apply 명령을 실행하면 Terraform이 실시간으로 상태
를 유지하는 데 사용하는 terraform.tfstate 파일을 생성하는 것을 볼 수 있습니다. 더 좋거나 더 나쁘면 
CloudFormation보다 Terraform의 상태를 훨씬 더 잘 제어할 수 있습니다. 상태 파일을 직접 업데이트
해서는 안 되지만 배포 간에 상태를 업데이트하는 몇 가지 Terraform CLI 명령을 실행할 수 있습니다. 
예를 들어, terraform 가져오기를 사용하면 Terraform 외부에서 생성된 리소스를 배포 스택에 추가할 
수 있습니다. 반대로 terraform state rm을 실행하여 상태에서 리소스를 제거할 수 있습니다.

Terraform이 상태를 어딘가에 저장해야 한다는 사실은 CloudFormation에 적용되지 않는 또 다른 개념
인 백엔드를 초래합니다. Terraform 백엔드는 배포 후 Terraform 스택이 상태 파일을 저장하는 곳입니
다. 새 배포가 시작될 때 상태 파일을 찾을 것으로 예상되는 위치이기도 합니다. 위에서 설명한 대로 로
컬에서 스택을 실행할 때 최상위 로컬 디렉터리에 Terraform 상태 사본을 보관할 수 있습니다. 이를 로
컬 백엔드라고 합니다.

지속적 통합 및 지속적 배포(CI/CD) 환경을 위해 개발할 때 로컬 상태 파일은 일반적으로 버전 제어에
서 벗어나도록 .gitignore 파일에 포함됩니다. 그러면 파이프라인 내에 로컬 상태 파일이 없습니다. 제
대로 작동하려면 파이프라인 단계가 어딘가에서 올바른 상태 파일을 찾아야 합니다.  이러한 이유로 
Terraform 구성 파일에는 종종 백엔드 블록이 포함됩니다. 백엔드 블록은 Terraform 스택에 상태 파일
을 찾기 위해 자체 최상위 디렉터리 이외의 위치를 찾아야 함을 나타냅니다.

Terraform 백엔드는 Amazon S3 버킷, API 엔드포인트 또는 원격 Terraform 워크스페이스 등 거의 모
든 곳에 위치할 수 있습니다. 다음은 Amazon S3 버킷에 저장된 Terraform 백엔드의 예입니다.

terraform { 
  backend "s3" { 
    bucket = "my-s3-bucket" 
    key    = "state-file-folder" 
    region = "us-east-1" 
  }
}

Terraform 구성 파일 내에 민감한 정보가 저장되지 않도록 백엔드는 부분 구성도 지원합니다. 이전 예
제에서는 버킷에 액세스하는 데 필요한 자격 증명이 구성에 없습니다. 자격 증명은 환경 변수 또는와 
같은 다른 수단을 사용하여 얻을 수 있습니다 AWS Secrets Manager. 자세한 내용은 AWS Secrets 
Manager 및 HashiCorp Terraform을 사용하여 민감한 데이터 보안을 참조하세요.

일반적인 백엔드 시나리오는 로컬 환경에서 테스트 목적으로 사용되는 로컬 백엔드입니다.
terraform.tfstate 파일은 원격 리포지토리로 푸시되지 않도록 .gitignore 파일에 포함됩니다. 그런 다음 
CI/CD 파이프라인 내의 각 환경은 자체 백엔드를 유지합니다. 이 시나리오에서는 여러 개발자가이 원
격 상태에 액세스할 수 있으므로 상태 파일의 무결성을 보호해야 합니다. 여러 배포가 동시에 실행되고 
상태를 업데이트하는 경우 상태 파일이 손상될 수 있습니다. 따라서 로컬이 아닌 백엔드가 있는 상황에
서는 일반적으로 배포 중에 상태 파일이 잠깁니다.

테라폼 데이터 소스 이해
배포 스택이 기존 리소스의 데이터에 의존하는 것은 매우 일반적입니다. 대부분의 IaC 도구에는 다
른 프로세스에서 만든 리소스를 가져오는 방법이 있습니다. 이렇게 가져온 리소스는 대개 읽기 전
용이며 (IAM 역할은 예외임) 스택 내 리소스에 필요한 데이터에 액세스하는 데 사용됩니다. AWS 
CloudFormation 리소스를 가져올 수 있지만, 를 보면 이 아이디어를 더 잘 설명할 수 있습니다. AWS 
Cloud Development Kit (AWS CDK)

개발자가 기존 프로그래밍 언어를 사용하여 CloudFormation 템플릿을 생성하는 AWS CDK 데 도움이 
됩니다. AWS CDK 작업의 최종 결과는 가져온 리소스입니다 CloudFormation. 그러나 와 함께 사용되
는 구문을 AWS CDK 사용하면 Terraform과 더 쉽게 비교할 수 있습니다. 다음은 를 사용하여 리소스를 
가져오는 예제입니다. AWS CDK

const importedBucket: IBucket = Bucket.fromBucketAttributes( 
    scope, 
    "imported-bucket", 
    { 
        bucketName: "My_S3_Bucket" 
    }
);

가져온 리소스는 일반적으로 같은 종류의 새 리소스를 만들 때 사용하는 것과 동일한 클래스에서 정적 
메서드를 호출하여 만들어집니다. 를 new Bucket(... 호출하면 새 리소스가 생성되고, 호출하면 기
존 리소스를 Bucket.fromBucketAttributes(... 가져옵니다. 올바른 버킷을 찾을 AWS CDK 수 
있도록 버킷 속성의 하위 집합을 함수에 전달합니다. 하지만 또 다른 차이점은 새 버킷을 만들면 모든 
속성과 메서드를 사용할 수 있는 Bucket 클래스의 전체 인스턴스가 반환된다는 것입니다. 리소스를 
가져오면 Bucket 반드시 있어야 하는 속성만 포함하는 IBucket 유형인 an이 반환됩니다. 외부 스택
에서 리소스를 가져올 수는 있지만 리소스로 수행할 수 있는 작업은 제한적입니다.

Terraform에서는 데이터 소스를 사용하여 유사한 목표를 달성합니다. 대부분의 정의된 Terraform 리소
스에는 함께 사용할 수 있는 데이터 소스가 함께 제공됩니다. 다음은 Terraform S3 버킷 리소스와 해당 
데이터 소스의 예입니다.

# S3 Bucket resource:
resource "aws_s3_bucket" "My_S3_Bucket" { 
  bucket = "My_S3_Bucket"
}

# S3 Bucket data source:

data "aws_s3_bucket" "My_S3_Bucket" { 
  bucket = "My_S3_Bucket"
}

이 두 항목의 유일한 차이점은 이름 접두사입니다. 데이터 원본 설명서에서 볼 수 있듯이 데이터 원본
에 전달할 수 있는 매개 변수는 리소스보다 적습니다. 이는 리소스가 이러한 파라미터를 사용하여 새 
S3 버킷의 모든 속성을 선언하는 반면, 데이터 원본에는 기존 리소스의 데이터를 고유하게 식별하고 
가져오기에 충분한 정보만 필요하기 때문입니다.

Terraform 리소스와 데이터 소스 구문 간의 유사성은 편리할 수 있지만 문제가 될 수도 있습니다. 
초보 Terraform 개발자는 구성에서 실수로 리소스 대신 데이터 소스를 사용하는 경우가 많습니다. 
Terraform 데이터 소스는 항상 읽기 전용입니다. 읽기 작업 (예: 다른 리소스에 ID 이름 제공) 에 해당 
리소스 대신 사용할 수 있습니다. 하지만 기본 리소스의 일부 측면을 근본적으로 변경하는 쓰기 작업에
는 사용할 수 없습니다. 이러한 이유로 Terraform 데이터 소스는 기본 리소스의 복제된 버전이라고 생
각할 수 있습니다.

이전 AWS CDK iBucket 예제와 마찬가지로 데이터 소스는 읽기 전용 시나리오에 유용합니다. 기존 리
소스에서 데이터를 가져와야 하지만 해당 리소스를 스택 내에 유지할 필요가 없는 경우 데이터 소스를 
사용하십시오. 계정의 기본 VPC를 사용하는 Amazon EC2 인스턴스를 생성하는 경우를 좋은 예로 들 
수 있습니다. VPC는 이미 존재하므로 데이터를 가져오기만 하면 됩니다. 다음 코드 샘플은 데이터를 
사용하여 대상 VPC를 식별하는 방법을 보여줍니다.

data "aws_vpc" "default" { 
  default = true
}

resource "aws_instance" "instance1" { 
  ami           = "ami-123456" 
  instance_type = "t2.micro" 
  subnet_id     = data.aws_vpc.default.main_route_table_id
}

Terraform 변수, 로컬 값 및 출력 이해
변수는 코드 블록 내의 자리 표시자를 허용하여 코드 유연성을 향상시킵니다. 변수는 코드를 재사용할 
때마다 서로 다른 값을 나타낼 수 있습니다. Terraform은 모듈식 범위에 따라 변수 유형을 구분합니다. 
입력 변수는 모듈에 삽입할 수 있는 외부 값이고, 출력 값은 외부에서 공유할 수 있는 내부 값이며, 로컬 
값은 항상 원래 범위 내에 유지됩니다.

변수
AWS CloudFormation 는 파라미터를 사용하여 한 스택 배포에서 다음 스택 배포로 설정하고 재설정할 
수 있는 사용자 지정 값을 나타냅니다. 마찬가지로 Terraform은 입력 변수 또는 변수를 사용합니다. 변
수는 Terraform 구성 파일의 모든 위치에서 선언할 수 있으며 일반적으로 필수 데이터 유형 또는 기본
값으로 선언됩니다. 다음 세 표현식 모두 유효한 Terraform 변수 선언입니다.

variable "thing_i_made_up" { 
  type = string
}

variable "random_number" { 
  default = 5
}

variable "dogs" { 
  type = list(object({ 
    name  = string 
    breed = string 
  })) 

  default = [ 
    { 
      name  = "Sparky", 
      breed = "poodle" 
    } 
  ]
}

구성 내에서 Sparky의 품종에 액세스하려면 변수를 사용합니다var.dogs[0].breed. 변수에 기본값
이 없고 null로 분류되지 않는 경우 각 배포에 대해 변수 값을 설정해야 합니다. 그렇지 않으면 변수에 
대한 새 값을 설정하는 것은 선택 사항입니다. 루트 모듈에서는 명령줄, 환경 변수 또는 terraform.tfvars 
파일에서 현재 변수 값을 설정할 수 있습니다. https://developer.hashicorp.com/terraform/language/ 

values/variables#variable-definitions-tfvars-files 다음 예제에서는 모듈의 최상위 디렉터리에 저장된
terraform.tfvars 파일에 변수 값을 입력하는 방법을 보여줍니다.

# terraform.tfvars
dogs = [ 
   {  
      name  = "Sparky",  
      breed = "poodle"  
   }, 
   {  
      name  = "Fluffy",  
      breed = "chihuahua"  
   }
]

random_number = 7

thing_i_made_up = "Kabibble"

dogs이 예제 terraform.tfvars 파일의 값은 변수 선언의 기본값을 재정의합니다. 하위 모듈 내에서 변수
를 선언하는 경우 다음 예제와 같이 모듈 선언 블록 내에서 직접 변수 값을 설정할 수 있습니다.

module "my_custom_module" { 
  source        = "modulesource/custom" 
  version       = "0.0.1" 
  random_number = 8
}

변수를 선언할 때 사용할 수 있는 다른 인수는 다음과 같습니다.

• sensitive -이 값을 로 설정하면 변수 값이 Terraform 프로세스 출력에 노출되지 true 않습니다. 

• nullable -이 값을 로 설정true하면 변수에 값이 없을 수 있습니다. 이는 기본값이 설정되지 않은 
변수에 유용합니다.

• description - 스택의 메타데이터에 변수에 대한 설명을 추가합니다.

• validation - 변수에 대한 검증 규칙을 설정합니다.

Terraform 변수의 가장 편리한 측면 중 하나는 변수 선언 내에 하나 이상의 검증 객체를 추가하는 기능
입니다. 검증 객체를 사용하여 변수가 전달해야 하는 조건을 추가할 수 있습니다. 그렇지 않으면 배포
가 실패합니다. 조건을 위반할 때마다 표시되도록 사용자 지정 오류 메시지를 설정할 수도 있습니다.

예를 들어 팀 구성원이 실행할 Terraform 구성 파일을 설정합니다. 스택을 배포하기 전에 팀원은 중요
한 구성 값을 설정하기 위해 terraform.tfvars 파일을 생성해야 합니다. 이를 상기시키기 위해 다음과 같
은 작업을 수행할 수 있습니다.

variable "important_config_setting" { 
  type = string 

  validation { 
    condition     = length(var.important_config_setting) > 0 
    error_message = "Don't forget to create the terraform.tfvars file!" 
  } 

  validation { 
    condition     = substr(var.important_config_setting, 0, 7) == "prefix-" 
    error_message = "Remember that the value always needs to start with 'prefix-'" 
  }
}

이 예제와 같이 단일 변수 내에서 여러 조건을 설정할 수 있습니다. Terraform은 실패한 조건에 대한 오
류 메시지만 표시합니다. 이렇게 하면 변수 값에 모든 종류의 규칙을 적용할 수 있습니다. 변수 값이 파
이프라인 장애를 일으키는 경우 그 이유를 정확히 알 수 있습니다.

로컬 값

모듈 내에 별칭을 지정하려는 값이 있는 경우 업데이트되지 않는 기본 변수를 선언하지 말고 locals
키워드를 사용합니다. 이름에서 알 수 있듯이 locals 블록에는 해당 특정 모듈로 내부적으로 범위가 
지정된 용어가 포함됩니다. 리소스 이름에 사용할 변수 값에 접두사를 추가하는 등 문자열 값을 변환하
려는 경우 로컬 값을 사용하는 것이 좋습니다. 다음 예제와 같이 단일 locals 블록은 모듈의 모든 로
컬 값을 선언할 수 있습니다.

locals { 
  moduleName    = "My Module" 
  localConfigId = concat("prefix-", var.important_config_setting)
}

값에 액세스할 때 locals 키워드는와 같이 단수로 바뀝니다local.LocalConfigId.

로컬 값 22

출력 값

Terraform 입력 변수가 CloudFormation 파라미터와 같은 경우 Terraform 출력 값이  CloudFormation 
출력과 같다고 말할 수 있습니다. 둘 다 배포 스택 내에서 값을 노출하는 데 사용됩니다. 그러나 
Terraform 모듈은 도구의 구조에 더 세분화되어 있으므로 Terraform 출력 값은 모듈 내의 값을 상위 모
듈 또는 다른 하위 모듈에 노출하는 데도 사용됩니다. 이러한 모듈이 모두 동일한 배포 스택 내에 있더
라도 마찬가지입니다. 두 개의 사용자 지정 모듈을 빌드하고 첫 번째 모듈이 두 번째 모듈의 ID 값에 액
세스해야 하는 경우 두 번째 모듈에 다음 output 블록을 추가해야 합니다.

output "module_id" { 
  value = local.module_id
}
Then in the first module you could use it like this:
module "first_module" { 
  source = "path/to/first/module"
}

resource "example_resource" "example_resource_name" { 
  module_id = module.first_module.module_id
}

Terraform 출력 값은 동일한 스택 내에서 사용할 수 있으므로 output 블록의 sensitive 속성
을 사용하여 스택 출력에 값이 표시되지 않도록 할 수도 있습니다. 또한 output 블록은 변수가
precondition 블록을 사용하는 것과 동일한 방식으로 validation 블록을 사용하여 변수가 특정 
규칙 세트를 따르도록 할 수 있습니다. 이렇게 하면 배포를 진행하기 전에 모듈 내의 모든 값이 예상대
로 존재하는지 확인할 수 있습니다.

output "important_config_setting" { 
  value = var.important_config_setting 

  precondition { 
    condition     = length(var.important_config_setting) > 0 
    error_message = "You forgot to create the terraform.tfvars file again." 
  }
}

Terraform 함수, 표현식 및 메타 인수 이해
일반적인 프로그래밍 언어보다는 선언적 구성 파일을 사용하는 IaC 도구에 대한 비판 중 하나는 사용
자 지정 프로그래밍 로직을 구현하기가 더 어렵다는 것입니다. Terraform 구성에서이 문제는 함수, 표
현식 및 메타 인수를 사용하여 해결됩니다.

함수
코드를 사용하여 인프라를 프로비저닝할 때 얻을 수 있는 큰 이점 중 하나는 공통 워크플로를 
저장 하고 반복해서 재사용하여 매번 다른 인수를 전달하는 것입니다. Terraform 함수는 AWS 
CloudFormation 내장 함수와 유사하지만 구문은 함수가 프로그래밍 언어로 호출되는 방식과 더 유사
합니다. 이 가이드의 예제에서 하위 문자열, concat,length 및 base64decode와 같은 일부 Terraform 함
수를 이미 발견했을 수 있습니다. 내장 함수가 있는 CloudFormation과 마찬가지로 Terraform에는 구성
에 사용할 수 있는 일련의 내장 함수가 있습니다. 예를 들어 특정 리소스 속성이 파일에 직접 붙여넣는 
데 비효율적인 매우 큰 JSON 객체를 사용하는 경우 객체를 .json 파일에 넣고 Terraform 함수를 사용
하여 액세스할 수 있습니다. 다음 예제에서 file 함수는 파일의 내용을 문자열 형식으로 반환한 다음
jsondecode 이를 객체 유형으로 변환합니다.

resource "example_resource" "example_resource_name" { 
  json_object = jsondecode(file("/path/to/file.json"))
}

Expressions
또한 Terraform은 보다 전통적인 삼원 연산자 구문을 사용한다는 점을 제외하면 CloudFormation
condition 함수와 유사한 조건식도 허용합니다. 다음 예제에서는 두 표현식이 정확히 동일한 결과를 
반환합니다. 두 번째 예는 Terraform이 스플랫 표현식을 호출하는 것입니다. 별표로 인해 Terraform은 
각 항목의 id 속성만 사용하여 목록을 반복하고 새 목록을 생성합니다.

resource "example_resource" "example_resource_name" { 
  boolean_value  = var.value ? true : false 
  numeric_value  = var.value > 0 ? 1 : 0 
  string_value   = var.value == "change_me" ? "New value" : var.value 
  string_value_2 = var.value != "change_me" ? var.value : "New value"
}
There are two ways to express for loops in a Terraform configuration:
resource "example_resource" "example_resource_name" { 
  list_value   = [for object in var.ids : object.id] 

  list_value_2 = var.ids[*].id
}

메타 인수

이전 코드 예제에서 list_value 및 list_value_2를 인수라고 합니다. 이러한 메타 인수 중 일부는 
이미 잘 알고 있을 수 있습니다. 또한 Terraform에는 인수처럼 작동하지만 몇 가지 추가 기능이 있는 몇 
가지 메타 인수가 있습니다.

• 메타 인수의 depends_on은 CloudFormation DependsOn 속성과 매우 유사합니다.

• 공급자 메타 인수를 사용하면 여러 공급자 구성을 한 번에 사용할 수 있습니다.

• 수명 주기 메타 인수를 사용하면 CloudFormation의 제거 및 삭제 정책과 마찬가지로 리소스 설정을 
사용자 지정할 수 있습니다.

다른 메타 인수를 사용하면 함수 및 표현식 기능을 리소스에 직접 추가할 수 있습니다. 예를 들어 개수
https://developer.hashicorp.com/terraform/language/meta-arguments/count메타 인수는 여러 유사한 
리소스를 동시에 생성하는 데 유용한 메커니즘입니다. 다음 예제에서는 count메타 인수를 사용하지 
않고 두 개의 Amazon Elastic Container Service(Amazon EKS) 클러스터를 생성하는 방법을 보여줍니
다.

resource "aws_eks_cluster" "example_0" { 
  name     = "example_0" 
  role_arn = aws_iam_role.cluster_role.arn 
  vpc_config { 
    endpoint_private_access = true 
    endpoint_public_access  = true 
    subnet_ids              = var.subnet_ids[0] 
  }
}

resource "aws_eks_cluster" "example_1" { 
  name     = "example_1" 
  role_arn = aws_iam_role.cluster_role.arn 
  vpc_config { 
    endpoint_private_access = true 
    endpoint_public_access  = true 
    subnet_ids              = var.subnet_ids[1] 
  }
}

다음 예제에서는 count메타 인수를 사용하여 두 개의 Amazon EKS 클러스터를 생성하는 방법을 보여
줍니다.

resource "aws_eks_cluster" "clusters" { 
  count    = 2 
  name     = "cluster_${count.index}" 
  role_arn = aws_iam_role.cluster_role.arn 
  vpc_config { 
    endpoint_private_access = true 
    endpoint_public_access  = true 
    subnet_ids              = var.subnet_ids[count.index] 
  }
}

각에 단위 이름을 지정하기 위해에서 리소스 블록 내의 목록 인덱스에 액세스할 수 있습니
다count.index. 하지만 조금 더 복잡한 유사한 리소스를 여러 개 생성하려면 어떻게 해야 할까
요? 여기에서 for_each meta-argument가 입력됩니다. for_each 메타 인수는 숫자 대신 목록이나 
객체를 전달하는 점을 count제외하면와 매우 유사합니다. Terraform은 목록 또는 객체의 각 멤버
에 대해 새 리소스를 생성합니다. 루프 인덱스가 아닌 목록의 내용에 액세스할 수 있다는 count = 
length(list)점을 제외하면를 설정하는 경우와 비슷합니다.

이는 항목 목록 또는 단일 객체 모두에 적용됩니다. 다음 예제에서는 IDid-1로 id-0 및가 있는 두 개
의 리소스를 생성합니다 IDs.

variable "ids" { 
  default = [ 
    { id = "id-0" }, 
    { id = "id-1" }, 
  ]
}

resource "example_resource" "example_resource_name" { 
  # If your list fails, you might have to call "toset" on it to convert it to a set 
  for_each = toset(var.ids) 
  id       = each.value
}

다음 예제에서는 Sparky용 리소스, 푸들용 리소스, chihuahua용 리소스 등 두 개의 리소스도 생성합니
다.

variable "dogs" { 

  default = { 
    poodle    = "Sparky" 
    chihuahua = "Fluffy" 
  }
}

resource "example_resource" "example_resource_name" { 
  for_each = var.dogs 
  breed    = each.key 
  name     = each.value
}

count.index를 사용하여 개수로 루프 인덱스에 액세스할 수 있는 것처럼 각 객체를 사용하여 for_each 
루프에 있는 각 항목의 키와 값에 액세스할 수 있습니다. for_each는 목록과 객체를 모두 반복하기 때문
에 각 키와 값은 추적하기가 약간 혼란스러울 수 있습니다. 다음 표에는 for_each meta-argument를 사
용할 수 있는 다양한 방법과 각 반복 시 값을 참조하는 방법이 나와 있습니다.

예제 for_each 유형 첫 번째 반복 두 번째 반복

A ["poodle", 
 "chihuahua"]

each.key = 
 "poodle" 

each.value = 
 null

each.key = 
 "chihuahua"

each.value = 
 null

B [

{

type = 
 "poodle", 

name = "Sparky" 

},

{

type = "chihuahu 
a",

each.key = {

type = "poodle", 

name = "Sparky" 

}

each.value = 
 null

each.key = {

type = "chihuahu 
a",

name = "Fluffy" 

}

each.value = 
 null

예제 for_each 유형 첫 번째 반복 두 번째 반복

name = "Fluffy" 

}

]

C {

poodle = 
 "Sparky", 

chihuahua = 
 "Fluffy" 

}

each.key = 
 "poodle" 

each.value = 
 "Sparky"

each.key = 
 "chihuahua"

each.value = 
 "Fluffy"

D {

dogs = {

poodle = 
 "Sparky", 

chihuahua = 
 "Fluffy" 

},

cats = {

persian = 
 "Felix", 

burmese = 
 "Morris" 

}

}

each.key = "dogs"

each.value = {

poodle = 
 "Sparky", 

chihuahua = 
 "Fluffy" 

}

each.key = "cats"

each.value = {

persian = 
 "Felix", 

burmese = 
 "Morris" 

}

예제 for_each 유형 첫 번째 반복 두 번째 반복

E {

dogs = [

{

type = 
 "poodle", 

name = "Sparky" 

},

{

type = "chihuahu 
a",

name = "Fluffy" 

}

],

cats = [

{

type = "persian" 
,

name = "Felix" 

},

{

type = "burmese" 
,

name = "Morris" 

each.key = "dogs"

each.value = [

{

type = 
 "poodle", 

name = "Sparky" 

},

{

type = "chihuahu 
a",

name = "Fluffy" 

}

]

each.key = "cats"

each.value = [

{

type = "persian" 
,

name = "Felix" 

},

{

type = "burmese" 
,

name = "Morris" 

}

]

예제 for_each 유형 첫 번째 반복 두 번째 반복

}

]

}

 

따라서 var.animals가 E행과 같으면 다음 코드를 사용하여 동물당 하나의 리소스를 생성할 수 있습
니다.

resource "example_resource" "example_resource_name" { 
  for_each = var.animals 
  type     = each.key 
  breeds   = each.value[*].type 
  names    = each.value[*].name
}

또는 다음 코드를 사용하여 동물당 두 개의 리소스를 생성할 수 있습니다.

resource "example_resource" "example_resource_name" { 
  for_each = var.animals.dogs 
  type     = "dogs" 
  breeds   = each.value.type 
  names    = each.value.name
}

resource "example_resource" "example_resource_name" { 
  for_each = var.animals.cats 
  type     = "cats" 
  breeds   = each.value.type 
  names    = each.value.name
}

메타 인수 30

FAQ

CloudFormation 대신 Terraform을 사용해야 하는 경우는 언제인가
요?

일반적으로 워크로드가 주로 기반인 경우 AWS는 Terraform이 일치시킬 수 없는 수준의 기본 지원을 
AWS CloudFormation 제공합니다. 그러나 워크로드에 타사 프로세스가 상당히 많거나 여러 클라우드 
공급자 간에 분산되어 있는 경우 Terraform은 고려할 수 있는 도구입니다.

CloudFormation AWS CDK 대신를 사용해야 하는 경우는 언제인가
요?

를 사용할 때는 CloudFormation AWS Cloud Development Kit (AWS CDK)도 사용합니다. 를 AWS 
CDK 사용하면 공통 프로그래밍 언어를 사용하여 CloudFormation 템플릿을 생성할 수 있습니다. 에서 
AWS CDK 지원하는 프로그래밍 언어에 대한 경험이 있는 경우는 AWS CDK CloudFormation 템플릿
을 생성하는 데 필요한 시간을 줄일 수 있습니다.

Terraform 구성을 AWS CDK 생성하는와 같은 도구가 있나요?

에 비해 AWS CDKCDK for Terraform(CDKTF)은 동일한 구문 라이브러리를 사용하여 리소스를 프
로비저닝하고 동일한 jsii 엔진을 사용하여 여러 프로그래밍 언어를 지원합니다. 이를 사용하여가 
CloudFormation 템플릿을 생성하는 것과 동일한 방식으로 Terraform 구성을 AWS CDK 생성할 수 있
습니다.

Terraform에 대해 자세히 알아보려면 어떻게 해야 하나요?

고급 Terraform 개념에 대한 자세한 내용은 Terraform 설명서를 참조하세요. 또한 모든 주요 공급자 및 
오픈 소스 모듈의 구성 요소에 대해서도 설명합니다.

CloudFormation 대신 Terraform을 사용해야 하는 경우는 언제인가요? 31

관련 리소스

AWS 설명서

• AWS CDK 설명서

• AWS CloudFormation 설명서

• 테라폼: 비욘드 더 베이직 위드 AWS (AWS 블로그 포스트)

기타 리소스

• 테라폼용 CDK 문서

• Terraform 설명서

AWS 설명서 32

https://docs.aws.amazon.com/cdk/v2/guide/getting_started.html
https://docs.aws.amazon.com/cloudformation/
https://aws.amazon.com/blogs/apn/terraform-beyond-the-basics-with-aws/
https://developer.hashicorp.com/terraform/cdktf
https://developer.hashicorp.com/terraform/docs

부록: Terraform 속성 액세스 예제

Resource

resource "aws_s3_bucket" "myS3Bucket" { 
     bucket = "my-s3-bucket"
}

bucketName = aws_s3_bucket.myS3Bucket.bucket

데이터 소스

data "aws_s3_bucket" "myS3Bucket" { 
     bucket = "my-s3-bucket"
}

bucketName = data.aws_s3_bucket.myS3Bucket.bucket

모듈

module "eks" { 
    source = "terraform-aws-modules/eks/aws" 
    version = "20.2.1"
}

vpc_id = module.eks.vpc_id

변수

variable "my_variable" = { 
   default = "dog"
}

animalType = var.my_variable

Resource 33

로컬

locals { 
  type = "dog"
}

animalType = local.type

로컬 34

문서 기록
아래 표에 이 가이드의 주요 변경 사항이 설명되어 있습니다. 향후 업데이트에 대한 알림을 받으려면
RSS 피드를 구독하십시오.

변경 사항 설명 날짜

최초 게시 — 2024년 3월 29일

35

https://docs.aws.amazon.com/prescriptive-guidance/latest/getting-started-terraform/getting-started-terraform.rss


AWS 권장 가이드 Terraform 시작하기: AWS CDK 및 AWS CloudFormation 전문가를 위
한 지침

기계 번역으로 제공되는 번역입니다. 제공된 번역과 원본 영어의 내용이 상충하는 경우에는 영어 버전
이 우선합니다.

